Doppelpendel mit Reibung (g=l=1)

In[1]:=
```
draw[{p1_, p2_}] := Module[{r1, r2}, r1 = {Sin[p1], -Cos[p1]};
  r2 = r1 + {Sin[p2], -Cos[p2]};
  Graphics[{AbsolutePointSize[10.2], Point[{0, 0}],
    AbsolutePointSize[20.2], AbsoluteThickness[3], Line[{{0, 0}, r1}],
    Line[{r1, r2}], {Blue, Point[r1]}, {Red, Point[r2]}}]]
```

In[2]:=
```
ClearAll[equations, eq1, eq2];
equations[m1_, m2_, γ1_, γ2_] =
 Module[{r1, r2, tt, eq1, eq2}, r1[tt_] = {Sin[φ1[tt]], -Cos[φ1[tt]]};
  r2[tt_] = r1[tt] + {Sin[φ2[tt]], -Cos[φ2[tt]]};
  lagrangian = 1/2 m1 r1'[t].r1'[t] +
     1/2 m2 r2'[t].r2'[t] - m1 r1[t][[2]] - m2 r2[t][[2]] // FullSimplify;
  eq1 = D[D[lagrangian, φ1'[t]], t] == D[lagrangian, φ1[t]] -
     γ1 r1'[t].D[r1[t], φ1[t]] - γ2 r2'[t].D[r2[t], φ1[t]] // FullSimplify;
  eq2 = D[D[lagrangian, φ2'[t]], t] == D[lagrangian, φ2[t]] -
     γ1 r1'[t].D[r1[t], φ2[t]] - γ2 r2'[t].D[r2[t], φ2[t]] // FullSimplify;
  {eq1,
   eq2}];
```

In[3]:=
```
equations[m1, m2, γ1, γ2]
```

Out[3]=
$$\left\{\left(\gamma 1+\gamma 2\right)\varphi 1'[t]+\gamma 2\,\text{Cos}[\varphi 1[t]-\varphi 2[t]]\,\varphi 2'[t]+m2\,\text{Sin}[\varphi 1[t]-\varphi 2[t]]\,\varphi 2'[t]^2+\left(m1+m2\right)\left(\text{Sin}[\varphi 1[t]]+\varphi 1''[t]\right)+m2\,\text{Cos}[\varphi 1[t]-\varphi 2[t]]\,\varphi 2''[t]==0,\ \gamma 2\,\text{Cos}[\varphi 1[t]-\varphi 2[t]]\,\varphi 1'[t]+\gamma 2\,\varphi 2'[t]+m2\left(\text{Sin}[\varphi 2[t]]+\text{Cos}[\varphi 1[t]-\varphi 2[t]]\,\varphi 1''[t]+\varphi 2''[t]\right)==m2\,\text{Sin}[\varphi 1[t]-\varphi 2[t]]\,\varphi 1'[t]^2\right\}$$

In[4]:=
```
ClearAll[solve0, solve, energy, eq1, eq2];
solve0[m1_, m2_, γ1_, γ2_, φ10_, φ20_] := solve0[m1, m2, γ1, γ2, φ10, φ20] =
  Module[{tt, eq1, eq2}, {eq1, eq2} = equations[m1, m2, γ1, γ2] /. t → tt;
   NDSolve[{eq1, eq2, φ1[0] == φ10, φ2[0] == φ20, φ1'[0] == 0, φ2'[0] == 0},
     {φ1, φ2}, {tt, 0, 40}][[1]]]
solve[m1_, m2_, γ1_, γ2_, φ10_, φ20_][t_] := {φ1[t], φ2[t]} /.
  solve0[m1, m2, γ1, γ2, φ10, φ20]

energy[m1_, m2_, γ1_, γ2_, φ10_, φ20_][t_] :=
 Module[{r1, r2, s = solve0[m1, m2, γ1, γ2, φ10, φ20], tt},
  r1[tt_] = {Sin[φ1[tt]], -Cos[φ1[tt]]} /. s;
  r2[tt_] = r1[tt] + {Sin[φ2[tt]], -Cos[φ2[tt]]} /. s;
  1/2 m1 r1'[t].r1'[t] + 1/2 m2 r2'[t].r2'[t] + m1 r1[t][[2]] + m2 r2[t][[2]]]
```

In[19]:=
```
Animate[Show[draw[solve[10, 1, .1, .1, Pi / 2, Pi / 2][t]],
  PlotRange → {{-2.2, 2.2}, {-2.2, 1}}], {t, 0, 40, .1}]
```

Out[19]=



Beispiel : m1 = 10, m2 = 1, ändere Reibung (stark bei erstem, letztem, keinen)

In[20]:=
```
Plot[energy[10, 1, 0.1, 0.1, Pi / 2, Pi / 2][t], {t, 0, 40}, PlotRange → All]
```

Out[20]=